VELEUČILIŠTE
BALTAZAR
ZAPREŠIĆ

Alisa Bilal Zorić

RAD S MAXIMOM

VELEUČILIŠTE BALTAZAR ZAPREŠIĆ

Zaprešić, 2022.

Alisa Bilal Zorić
Rad s Maximom

Nakladnik
Veleučilište s pravom javnosti *Baltazar Zaprešić*

Za nakladnika
Ivan Ružić

Recenzenti
dr. sc. Vladimir Mateljan
dr. sc. Romana Capor Hrošik

Uredništvo
Ivan Ružić, glavni i odgovorni urednik
Gordana Šiber, izvršna urednica
Petra Popek Biškupec, članica
Ana Skledar Ćorluka, članica
Zlatko Rešetar, član

Lektor
Lektor d. o. o.

Grafički urednik
Damir Vidaković

Podatak o izdanju
1. internetsko izdanje

Datum objave na mreži
31. 3. 2022.

Dostupno na
https://www.bak.hr/hr/referada/knjiznica/e-knjiznica

ISBN
978-953-8037-24-5

Alisa Bilal Zorić

RAD S MAXIMOM

Veleučilište Baltazar Zaprešić

Zaprešić, 2022.

Sadržaj

Predgovor 1
1. Uvod 2
   1.1. Korisničko sučelje 2
2. Posebni znakovi i simboli 3
3. Pomoć 4
4. Konstante 5
5. Aritmetičko-logičke operacije 6
6. Funkcije i algebarski izrazi 9
7. Limesi, derivacije, integrali 12
   7.1. Limesi 12
   7.2. Derivacije 13
   7.3. Integrali 13
8. Suma i produkt 14
9. Grafovi 15
10. Matrice 18
   10.1. Unos matrica 18
   10.2. Operacije s matricama 19
   10.3. Funkcije za rad s matricama 21
11. Linearne jednadžbe i nejednadžbe 23
12. *For* petlja 25
13. Primjeri iz nastave 26
   13.1. Matrice 26
      13.1.1. Operacije s matricama 26
      13.1.2. Matrične jednadžbe 35
   13.2. Primjena matrica u ekonomiji 40
      13.2.1. *Input-output* analiza 40
   13.3. Limesi, derivacije, integrali 48
      13.3.1. Primjena derivacija i integrala u ekonomiji 53
   13.4. Površine 56
   13.5. Redovi 58
Literatura 64

Predgovor

Ovaj priručnik namijenjen je prvenstveno studentima Veleučilišta Baltazar Zaprešić, a može poslužiti i svim drugima zainteresiranima za rad u Maximi. Priručnik je podijeljen u dva osnovna dijela: *Općenito o radu u Maximi* i *Riješeni primjeri iz nastave*.

Riješeni primjeri iz nastave pojavljuju se u kolegijima Ekonomska matematika i Poslovna matematika na studiju Poslovanje i upravljanje te Matematika 1 i Matematika 2 na studiju Informacijske tehnologije.

U priručniku nisu navedena teoretska objašnjenja zadataka ni definicije, nego samo praktični rad u Maximi.

1. Uvod

Maxima je besplatni slobodni softverski paket namijenjen rješavanju matematičkih problema. Zasnovan je na verziji Macsyma iz 1982., a razvijen je na MIT-u. Besplatno preuzimanje pod uvjetima licencije GNU GPL dostupno je na https://maxima.sourceforge.io/
Preuzimanje Maxime dostupno je u raznim operacijskim sustavima, uključujući MS Windows, Mac OSX i Linux. U ovom je priručniku korištena verzija Maxime 20.06.6.

1.1. Korisničko sučelje

Izgled sučelja WxMaxima prikazan je na slici 1.



Slika 1. Sučelje WxMaxima

Sučelje se sastoji od alatne trake i bloka za pisanje. Iz alatne trake mogu se pozivati razne funkcije. Funkcije se mogu upotrebljavati i upisom određene naredbe u blok za pisanje. Maximina datoteka snima se na standardan način: *Datoteka* → *Snimi* ili *Snimi kao*. Snimljena datoteka ima nastavak `.wxmx`. Otvaranje snimljene Maximine datoteke također je standardno, *Datoteka* → *Otvori*.

2. Posebni znakovi i simboli

Znak za završetak naredbe jest točka sa zarezom (;) i potom **`Enter`**. Ako želite da se naredba odmah izvrši, potrebno je istodobno pritisnuti **`Shift i Enter`**. Ako se na kraju naredbe ne napiše točka sa zarezom, pritiskom na tipke **`Shift i Enter`** Maxima će ga sama dodati.
Sve što se upiše između znakova **/*** i ***/** Maxima smatra komentarom.

Maxima razlikuje VELIKA i mala slova!

Za pridruživanje vrijednosti varijabli upotrebljava se dvotočka (:), a ne znak jednakosti (=). Znak jednakosti upotrebljava se za prikaz vrijednosti. Za definiranje funkcija upotrebljavamo dvotočku i znak jednakosti (:=).

Operator	Opis
`:`	operator pridruživanja vrijednosti (=)
`:=`	operator za definiranje funkcija
`kill(x)`	briše sve poveznice od *x*
`kill(all)`	briše sve poveznice (engl. *clear memory*)
`/* */`	sve između tretira se kao komentar

```
→     /* komentar*/;

(%i17)x;
(%o17)   x

(%i20)x:7;
(%o20)   7

(%i21)x;
(%o21)   7

(%i22)f(x):=2·x+5;
(%o22)   f (x):=2 x+5

(%i23)f(3);
(%o23)   11
```

Slika 2. Primjer pridruživanja u Maximi

3. Pomoć

Ako nas zanima što radi neka funkcija u Maximi ili želimo saznati njezin opis, dovoljno je upisati ***??****funkcija* ili ***describe****(funkcija)*.

```
(%i24)??sqrt;
        0: isqrt  (Functions and Variables for Number Theory)
        1: sqrt  (Root Exponential and Logarithmic Functions)
        2: sqrtdenest  (Functions and Variables for Expressions)
        3: sqrtdispflag  (Functions and Variables for Display)
Enter space-separated numbers, `all' or `none':0;
        -- Function: isqrt (<x>)
            Returns the "integer square root" of the absolute value of <x>,
            which is an integer.
(%o24)  true

(%i35)describe(ratsimp);
        0: ratsimp  (Functions and Variables for Polynomials)
        1: ratsimp <1>  (Functions and Variables for Polynomials)
Enter space-separated numbers, `all' or `none':0;
        -- Function: ratsimp (<expr>)
        -- Function: ratsimp (<expr>, <x_1>, ..., <x_n>)
            Simplifies the expression <expr> and all of its subexpressions,
            including the arguments to non-rational functions.  The result is
            returned as the quotient of two polynomials in a recursive form,
            that is, the coefficients of the main variable are polynomials in
            the other variables.  Variables may include non-rational functions
            (e.g., 'sin (x^2 + 1)') and the arguments to any such functions are
            also rationally simplified.
```

Slika 3. Primjer poziva pomoći u Maximi

4. Konstante

Konstante su veličine koje se ne mijenjaju tijekom izvođenja programa. Posebne konstante definiraju se znakom postotka ispred naziva. U nastavku su navedene konstante koje su unaprijed ugrađene u Maximu.

Konstanta	Opis konstante
`%e`	Eulerova konstanta $e = 2{,}718\,281\,828\,459\,04$
`%pi`	$\pi = 3{,}141\,592\,653\,589\,793$
`%i`	imaginarna jedinica $i = \sqrt{-1}$
`Inf`	beskonačno ∞
`minf`	minus beskonačno –∞

```
(%i8) pi:%pi;
(%o8) π

(%i9) float(pi);
(%o9) 3.141592653589793

(%i10)e:%e;
(%o10) %e

(%i11)float(e);
(%o11) 2.718281828459045

(%i12)%i^2;
(%o12) -1
```

Slika 4. Primjer ugrađenih konstanti u Maximi

5. Aritmetičko-logičke operacije

Aritmetička operacija	Opis
`+`	zbrajanje
`-`	oduzimanje
`*`	množenje
`/`	dijeljenje
`^`	potenciranje
`sqrt(x) ili x^(1/2)`	drugi korijen
`x^(1/n)`	*n*-ti korijen
`float(x)`	pretvara *x* u decimalni oblik

Napomena: Novije verzije Maxime zvjezdicu (*) kod množenja pretvaraju u točku (vidi primjer u nastavku).

```
(%i1) 2+3;
(%o1) 5

(%i2) 56-8;
(%o2) 48

(%i3) 9·3;
(%o3) 27

(%i4) 244/4;
(%o4) 61

(%i5) 3^5;
(%o5) 243

(%i6) sqrt(9);
(%o6) 3

(%i7) 27^(1/3);
(%o7) 3
```

Slika 5. Primjeri algebarskih operacija u Maximi

Maxima sve rezultate prikazuje što točnije može, što znači da krati razlomke, pojednostavnjuje aritmetički izraz, prikazuje posebne navedene konstante, ali rezultat ne prikazuje u decimalnom obliku. Da bismo dobili rezultat u decimalnom obliku, potrebno je pozvati naredbu *float(%)*, koja prethodno izračunani rezultat pretvara u decimalni oblik.



Slika 6. Primjer rezultata u Maximi

Operator	**Opis**
<	manje od
<=	manje od ili jednako
>	veće od
>=	veće od ili jednako
=	jednakost
equal(x,y)	jesu li *x* i *y* ekvivalentni
and	logički operator „i”
or	logički operator „ili”
not	negacija
true	istina
false	laž

unknown	neodređena logička tvrdnja
is(x)	određuje je li *x* istina, laž ili neodređeno

```
(%i17)is(4>7);
(%o17)  false

(%i20)3>2 or 5<3;
(%o20)  true

(%i21)is(x>8);
(%o21)  unknown

(%i22)x:(a+b)·(a-b);
(%o22)  (a-b) (b+a)

(%i23)y:a^2-b^2;
         2   2
(%o23)  a  -b

(%i24)is(x=y);
(%o24)  false

(%i25)is(equal(x,y));
(%o25)  true
```

Slika 7. Primjer logičkih operatora

6. Funkcije i algebarski izrazi

`Funkcija`	Opis
`abs(x)`	apsolutna vrijednost od *x*
`log(x)`	prirodni logaritam (baza e, $\ln(x)$)
`exp(x) ili %e^x`	eksponencijalna funkcija od *x* (e^x)
`sin(x)`	sinus od *x*
`cos(x)`	kosinus od *x*
`tan(x)`	tangens od *x*
`cot(x)`	kotangens od *x*

Napomena: Kod trigonometrijskih funkcija *x* je u radijanima (pravi je kut $\frac{\pi}{2}$). Da bi se stupnjevi *s* pretvorili u radijane *r*, potrebno ih je pomnožiti s $\frac{\pi}{180}$ ($r = s\frac{\pi}{180}$).

Log(x) je u Maximi ln(*x*). Logaritam po bazi 10 možemo dobiti kao $log_{10}(x) = \frac{\log(x)}{\log(10)}$.

```
(%i13)cos(%pi);
(%o13)  -1

(%i14)abs(-9);
(%o14)  9

(%i15)log(%e);
(%o15)  1

(%i16)exp(4);
          4
(%o16)  %e
```

Slika 8. Primjeri funkcija u Maximi

Funkcije koje ćemo sami definirati označavat ćemo malim slovima počevši od *f* (*f*(*x*), *g*(*x*), *h*(*x*) i tako dalje).

```
(%i36) f(x):=2·x^2-4·x+7;
(%o36) f(x):=2 x^2 -4 x+7

(%i38) f(5);
(%o38) 37

(%i37) g(x):=(x^3-7)/(2·x-5)^(1/3);
(%o37) g(x):=(x^3 -7)/(2 x-5)^(1/3)

(%i39) g(3);
(%o39) 20
```

Slika 9. Primjer definiranja funkcija

Funkcija	**Opis**
`expand(izraz)`	raspisuje izraz (pomnoži sve zagrade)
`factor(izraz)`	faktorizira izraz
`num(izraz)`	vraća brojnik izraza
`denom(izraz)`	vraća nazivnik izraza
`combine(izraz)`	svodi izraz na zajednički nazivnik
`if(uvjet) then(vrijednost1) else(vrijednost2)`	uvjetno definiranje

```
(%i28) aps(x):=if(x>=0) then x else -x;
(%o28) aps(x):=if x≥0 then x else -x

(%i29) aps(-4);
(%o29) 4

(%i30) aps(0);
(%o30) 0

(%i31) aps(20);
(%o31) 20
```

Slika 10. Primjer uvjetno definirane funkcije



(%i1) f(x):=(2·x+3)^3;

(%o1) $f(x) := (2x+3)^3$

(%i2) expand(f(x));

(%o2) $8x^3 + 36x^2 + 54x + 27$

(%i3) factor(%);

(%o3) $(2x+3)^3$

(%i5) g(x):=(2·x+7)^2/(x^2-9);

(%o5) $g(x) := \frac{(2x+7)^2}{x^2-9}$

(%i7) expand(g(x));

(%o7) $\frac{4x^2}{x^2-9} + \frac{28x}{x^2-9} + \frac{49}{x^2-9}$

(%i8) num(g(x));

(%o8) $(2x+7)^2$

(%i10) denom(g(x));

(%o10) x^2-9

(%i1) combine(x/3+y/4);

(%o1) $\frac{3y+4x}{12}$

Slika 11. Primjer manipulacije izrazima

7. Limesi, derivacije, integrali

7.1. Limesi

Funkcija	Opis
`limit(funkcija,varijabla, vrijednost)`	limes funkcije kad varijabla teži vrijednosti
`limit(funkcija,varijabla, vrijednost,smjer)`	limes funkcije kad varijabla teži vrijednosti iz smjera (slijeva ili zdesna)

```
(%i5) limit(1/x,x,inf);
(%o5) 0

(%i6) limit(1/x,x,0);
(%o6) infinity

(%i8) limit((%e^x-%e^-x)/(sin(x)),x,0);
(%o8) 2

(%i18) limit(x/abs(x),x,0,minus);
(%o18) -1

(%i19) limit(x/abs(x),x,0,plus);
(%o19) 1

(%i20) limit((x^2-49)/(2-(x-3)^(1/2)),x,7);
(%o20) -56
```

Slika 12. Primjer računanja limesa

7.2. Derivacije

Funkcija	Opis
`diff(funkcija,varijabla)`	derivacija funkcije po varijabli
`diff(funkcija,varijabla,n)`	*n*-ta derivacija funkcije po varijabli

```
(%i13) diff(x^5-3·x^3+7·x-8,x);
```

$$\text{(\%o13)}\quad 5\,x^4-9\,x^2+7$$

```
(%i14) diff(x^5-3·x^3+7·x-8,x,3);
```

$$\text{(\%o14)}\quad 60\,x^2-18$$

Slika 13. Primjer derivacija

7.3. Integrali

Funkcija	Opis
`integrate(funkcija,varijabla)`	računa integral funkcije po varijabli
`integrate(funkcija,varijabla,a,b)`	računa određeni integral od *a* do *b*

```
(%i15) integrate(5·x^4-9·x^2+7,x);
```

$$\text{(\%o15)}\quad x^5-3\,x^3+7\,x$$

```
(%i16) integrate(-x^2+x+2,x,-1,2);
```

$$\text{(\%o16)}\quad \frac{9}{2}$$

Slika 14. Primjer računanja integrala

Napomena: Maxima u rezultatu ne prikazuje sva rješenja neodređenog integrala, nego daje samo jedno rješenje, što znači da sami moramo dodati + C u konačnom rezultatu.

8. Suma i produkt

Funkcija	Opis
sum(izraz,i_varijabla,od_1,do_n)	računa zbroj izraza po *i*_varijabli od_1 do_*n*; do_*n* može biti i beskonačno (inf)
simpsum	ako se eksplicitno navede iza sume, Maxima će pokušati izračunati pravu sumu
product(izraz,i_varijabla,od_1,do_n)	računa produkt izraza po *i*_varijabli od_1 do_*n*; do_*n* može biti i beskonačno (inf)
simpproduct	ako se eksplicitno navede iza produkta, Maxima će pokušati izračunati pravi produkt

```
(%i22) sum(i,i,1,100);
(%o22) 5050
```

```
(%i3) sum(1/(i^2),i,1,inf),simpsum;
```

(%o3) $\frac{\pi^2}{6}$

```
(%i11) sum(1/4^i,i,1,inf);
```

(%o11) $\sum_{i=1}^{\infty}\left(\frac{1}{4^i}\right)$

```
(%i1) sum(1/4^i,i,1,inf),simpsum;
```

(%o1) $\frac{1}{3}$

```
(%i5) product(i^2,i,1,10);
(%o5) 13168189440000
```

```
(%i12) product(1/(i^2),i,1,inf);
```

(%o12) $\prod_{i=1}^{\infty}\left(\frac{1}{i^2}\right)$

```
(%i10) product(1/(i^2),i,1,10);
```

(%o10) $\frac{1}{13168189440000}$

Slika 15. Primjer sume i produkta prvih n brojeva

9. Grafovi

Crtanje dvodimenzionalnih grafova u Maximi postiže se naredbom ***Plot2d***`([f(x)],[x, x_min,x_max],[y,y_min,y_max]),` za os *y* parametri su opcionalni.
Ova naredba crta graf u novom prozoru te se taj prozor mora zatvoriti kako bi se moglo nastaviti raditi u Maximi. Ako želimo nacrtati graf u aktivnom prozoru Maxime, to je moguće naredbom ***wxplot2d*** s istim parametrima.
Slično se crtaju i trodimenzionalni grafovi u Maximi. Naredba je ***plot3d*** odnosno ***wxplot3d*** ako se želi prikazati u aktivnom prozoru.

Naredba	Opis
plot2d([funkcije**],[**x,x_min,x_max**],** ***[y,y_min,y_max])***	crta dvodimenzionalni graf funkcija od *x*_min do *x*_max i od *y*_min do *y*_max u novom prozoru; raspon za *x* je obvezan, a ako se ne navede raspon za *y*, Maxima ga automatski odredi
wxplot2d([funkcije**],[**x,x_min,x_max**],** ***[y,y_min,y_max])***	crta dvodimenzionalni graf funkcija od *x*_min do *x*_max i od *y*_min do *y*_max u glavnom prozoru; raspon za *x* je obvezan, a ako se ne navede raspon za *y*, Maxima ga automatski odredi
plot3d([funkcije**],[**x,x_min,x_max**],** ***[y,y_min,y_max])***	crta trodimenzionalni graf funkcija od *x*_min do *x*_max i od *y*_min do *y*_max u novom prozoru; raspon za *x* je obvezan, a ako se ne navede raspon za *y*, Maxima ga automatski odredi
wxplot3d([funkcije**],[**x,x_min,x_max**],** ***[y,y_min,y_max])***	crta trodimenzionalni graf funkcija od *x*_min do *x*_max i od *y*_min do *y*_max u glavnom prozoru; raspon za *x* je obvezan, a ako se ne navede raspon za *y*, Maxima ga automatski odredi

Napomena: Navedene naredbe mogu istodobno nacrtati grafove više funkcija. Također, mogu se dodati i dodatni parametri. Funkcije se mogu zadati implicitno, eksplicitno i parametarski. Za više detalja ukucati ***??plot2d*** ili ***??plot3d***.



Slika 16. Primjer crtanja dvodimenzionalnih grafova



Slika 17. Primjer trodimenzionalnog grafa



Slika 18. Primjer dodatnih opcija za grafički prikaz

10. Matrice

10.1. Unos matrica

Najjednostavniji način za unošenje matrica jest preko dijaloškog okvira odabirom *Algebra* → *Enter Matrix*.





Tu definiramo koliko će naša matrica imati redaka (*Rows*) i Stupaca (*Columns*) i damo joj ime (*Name*).
Matrice obično označavamo velikim slovima abecede, matrica *A*, *B*, *C* itd. Pritiskom na Enter otvara se dijaloški okvir za unos elemenata matrice.



Isto se dobije naredbom **`matrix(`**`[elementi prvog retka]`**`,`**`[elementi drugog retka]`**`,`**`[elementi trećeg retka]`**`)`**. Elemente retka odvajamo zarezom.



Slika 19. Primjer unosa matrice

10.2. Operacije s matricama

Operacija	**Opis**
`A+B`	zbrajanje matrica A i B
`A-B`	oduzimanje matrica A i B
`n*A`	množenje matrice A skalarom n
`A.B`	množenje matrica A i B
`A^^n`	n-ta potencija matrice A
`A[i,j]`	vraća element koji se nalazi na presjeku i-tog retka i j-tog stupca

(%i1) A:matrix([2,1],[-1,3]);

(%o1) $\begin{pmatrix} 2 & 1 \\ -1 & 3 \end{pmatrix}$

(%i2) B:matrix([1,3],[2,-1]);

(%o2) $\begin{pmatrix} 1 & 3 \\ 2 & -1 \end{pmatrix}$

(%i3) A+B;

(%o3) $\begin{pmatrix} 3 & 4 \\ 1 & 2 \end{pmatrix}$

(%i4) 2·A;

(%o4) $\begin{pmatrix} 4 & 2 \\ -2 & 6 \end{pmatrix}$

(%i5) A.B;

(%o5) $\begin{pmatrix} 4 & 5 \\ 5 & -6 \end{pmatrix}$

(%i6) A-B;

(%o6) $\begin{pmatrix} 1 & -2 \\ -3 & 4 \end{pmatrix}$

(%i7) A^^3;

(%o7) $\begin{pmatrix} 1 & 18 \\ -18 & 19 \end{pmatrix}$

(%i8) A[2,2];

(%o8) 3

Slika 20. Primjer operacija s matricama

Napomena: *A***B* pomnožit će odgovarajuće elemente, *A*^2 potencirat će svaki element matrice!

(%i9) A·B;

(%o9) $\begin{pmatrix} 2 & 3 \\ -2 & -3 \end{pmatrix}$

(%i10) A^2;

(%o10) $\begin{pmatrix} 4 & 1 \\ 1 & 9 \end{pmatrix}$

10.3. Funkcije za rad s matricama

Maxima ima široki raspon funkcija za manipuliranje matricama, a ovdje ćemo spomenuti samo neke koje ćemo kasnije upotrebljavati u zadacima.

Naredba	Opis
transpose(A)	transponirana matrica matrice *A*
determinant(A)	determinanta matrice *A*
invert(A) **ili** A^^(-1)	inverzna matrica matrice *A*
ident(n)	jedinična matrica reda *n*
genmatrix(A,m,n)	generira matricu *A* zadanu pravilom *A*[*i*,*j*] s *m* redaka i *n* stupaca



Slika 21. Primjer rada s matricama

```
(%i6) ident(3);
```

$$(\%o6)\quad \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
(%i1) A[i,j]:=i+j;
```

$$(\%o1)\quad A_{i,j} := i + j$$

```
(%i3) genmatrix(A,3,4);
```

$$(\%o3)\quad \begin{pmatrix} 2 & 3 & 4 & 5 \\ 3 & 4 & 5 & 6 \\ 4 & 5 & 6 & 7 \end{pmatrix}$$

Slika 22. Primjer generiranja matrice

11. Linearne jednadžbe i nejednadžbe

Sustav `n` linearnih jednadžbi s `n` nepoznanica možemo riješiti s pomoću matrica. Sustav možemo zapisati u matričnom obliku kao `A.x=b` pa množenjem lijeve i desne strane jednadžbe s inverzom od *A* s lijeve strane, dobivamo da je `x=A^(-1).b`.

```
(%i8) A: matrix([3,-4,-1], [5,-3,-4], [4,-2,3]);
```

(%o8) $\begin{pmatrix} 3 & -4 & -1 \\ 5 & -3 & -4 \\ 4 & -2 & 3 \end{pmatrix}$

```
(%i10) b: matrix([-2], [-2], [5]);
```

(%o10) $\begin{pmatrix} -2 \\ -2 \\ 5 \end{pmatrix}$

```
(%i11) determinant(A);
```

(%o11) 71

```
(%i15) x:invert(A).b;
```

(%o15) $\begin{pmatrix} 1 \\ 1 \\ 1 \end{pmatrix}$

```
(%i17) A.x=b;
```

(%o17) $\begin{pmatrix} -2 \\ -2 \\ 5 \end{pmatrix} = \begin{pmatrix} -2 \\ -2 \\ 5 \end{pmatrix}$

Slika 23. Primjer rješavanja linearnih jednadžbi

Jednadžbe se u Maximi rješavaju naredbom **solve([**jednadžbe**],[**nepoznanice**]).**

```
(%i3) solve([x^2-13·x+36=0],[x]);
(%o3) [x=9,x=4]

(%i4) solve([x^4-13·x^2+36=0],[x]);
(%o4) [x=-2,x=2,x=-3,x=3]

(%i2) solve([3·p1+3·p2-2·p3=-3,2·p1-3·p2+2·p3=8,
       -p1+2·p2+3·p3=21],[p1,p2,p3]);
(%o2) [[p1=1,p2=2,p3=6]]
```

Slika 24. Primjer rješavanja jednadžbi

Za rješavanje nejednadžbi potrebno je prvo učitati paket solve_rat ineq. To se radi naredbom **load** (load(solve_rat_ineq)) i tek nakon toga možemo rješavati nejednadžbe naredbom **solve_rat_ineq(**nejednadžba**).**

```
(%i16)load(solve_rat_ineq);
(%o16)
      C:/maxima-5.44.0/share/maxima/5.44.0/share/solve_rat_ineq/solve_rat_ineq.mac

(%i22)solve_rat_ineq((x+1)/(1-x)>=0);
(%o22) [[x≥-1,x<1]]

(%i23)solve_rat_ineq(2·x+5>=-x-7);
(%o23) [[x≥-4]]

(%i28)solve_rat_ineq(3/4·x+5<2·x);
(%o28) [[x>4]]
```

Slika 25. Primjer rješavanja nejednadžbi

12. *For* petlja

For petlja omogućava uzastopno ponavljanje određenog koda (tijelo petlje). *For* petlja primjenjuje se kada unaprijed znamo broj iteracija. Kod *for* petlje važno je na početku inicirati varijablu po kojoj npr. zbrajamo ili množimo (*S:0* ili *S:1*).

```
(%i1) S:0;
(%o1) 0

(%i2) for i:1 thru 100 do
      S:S+(1/(i^2-(1/4)));
(%o2) done

(%i3) S;
       400
(%o3)  ---
       201

(%i4) float(S);
(%o4) 1.990049751243781
```

Slika 26. Primjer for *petlje*

13. Primjeri iz nastave

13.1. Matrice

13.1.1. Operacije s matricama

Primjer 1. Za matrice $A = \begin{pmatrix} 1 & 5 & -2 \\ -3 & 0 & 2 \end{pmatrix}$, $B = \begin{pmatrix} -3 & 0 & 12 \\ 21 & 1 & 0 \end{pmatrix}$ i $C = \begin{pmatrix} 10 & 8 & 5 \\ -7 & -2 & 1 \end{pmatrix}$ izračunajte $2\,A + \frac{1}{3}B - C$.

Rješenje:

```
(%i33) A: matrix([1,5,-2], [-3,0,2]);
```

$$(\%o33) \quad \begin{pmatrix} 1 & 5 & -2 \\ -3 & 0 & 2 \end{pmatrix}$$

```
(%i34) B: matrix([-3,0,12], [21,1,0]);
```

$$(\%o34) \quad \begin{pmatrix} -3 & 0 & 12 \\ 21 & 1 & 0 \end{pmatrix}$$

```
(%i35) C: matrix([10,8,5], [-7,-2,1]);
```

$$(\%o35) \quad \begin{pmatrix} 10 & 8 & 5 \\ -7 & -2 & 1 \end{pmatrix}$$

```
(%i36) 2·A+(1/3)·B-C;
```

$$(\%o36) \quad \begin{pmatrix} -9 & 2 & -5 \\ 8 & \frac{7}{3} & 3 \end{pmatrix}$$

Primjer 2. Za matrice $A = \begin{pmatrix} 1 & 2 & 1 \\ 2 & 1 & 2 \\ 1 & 2 & 3 \end{pmatrix}$ i $B = \begin{pmatrix} 4 & 1 & 1 \\ -4 & 2 & 0 \\ 1 & 2 & 1 \end{pmatrix}$ izračunajte $A \cdot B$ i $B \cdot A$. Što zaključujete?

Rješenje:

```
(%i2) A: matrix([1,2,1], [2,1,2], [1,2,3]);
```

(%o2) $\begin{pmatrix} 1 & 2 & 1 \\ 2 & 1 & 2 \\ 1 & 2 & 3 \end{pmatrix}$

```
(%i3) B: matrix([4,1,1], [-4,2,0], [1,2,1]);
```

(%o3) $\begin{pmatrix} 4 & 1 & 1 \\ -4 & 2 & 0 \\ 1 & 2 & 1 \end{pmatrix}$

```
(%i4) A.B;
```

(%o4) $\begin{pmatrix} -3 & 7 & 2 \\ 6 & 8 & 4 \\ -1 & 11 & 4 \end{pmatrix}$

```
(%i5) B.A;
```

(%o5) $\begin{pmatrix} 7 & 11 & 9 \\ 0 & -6 & 0 \\ 6 & 6 & 8 \end{pmatrix}$

Zaključujemo da množenje matrica nije komutativno, odnosno da **ne vrijedi** $A \cdot B = B \cdot A$.

Primjer 3. Neka je $A = \begin{pmatrix} 2 & 1 \\ -3 & 2 \end{pmatrix}$. Izračunajte $A^3 - A + 3\,I$.

Rješenje:

```
(%i9) A: matrix([2,1], [-3,2]);
```

$$(\%o9)\quad \begin{pmatrix} 2 & 1 \\ -3 & 2 \end{pmatrix}$$

```
(%i7) I:ident(2);
```

$$(\%o7)\quad \begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

```
(%i8) A^^3-A+3·I;
```

$$(\%o8)\quad \begin{pmatrix} -9 & 8 \\ -24 & -9 \end{pmatrix}$$

Primjer 4: Za matricu $A = \begin{pmatrix} 1 & 2 & 3 \\ 2 & -1 & -1 \\ 1 & 3 & 4 \end{pmatrix}$ izračunajte determinantu i inverznu matricu.

Rješenje:

```
(%i10) A: matrix([1,2,3], [2,-1,-1], [1,3,4]);
```

(%o10) $\begin{pmatrix} 1 & 2 & 3 \\ 2 & -1 & -1 \\ 1 & 3 & 4 \end{pmatrix}$

```
(%i11) determinant(A);
```

(%o11) 2

```
(%i12) invert(A);
```

(%o12) $\begin{pmatrix} -\frac{1}{2} & \frac{1}{2} & \frac{1}{2} \\ -\frac{9}{2} & \frac{1}{2} & \frac{7}{2} \\ \frac{7}{2} & -\frac{1}{2} & -\frac{5}{2} \end{pmatrix}$

Primjer 5. Zadana je matrica $A = (i - j) \in M_4$. Ispitajte je li matrica A:

a) simetrična

b) antisimetrična.

Uputa: Promotrite matrice $C = A^{\tau} - A \; i \; D = A^{\tau} + A$.

Rješenje:

(%i1) `A[i,j]:=i-j;`

(%o1) $A_{i,j} := i - j$

(%i2) `A:genmatrix(A,4);`

(%o2) $\begin{pmatrix} 0 & -1 & -2 & -3 \\ 1 & 0 & -1 & -2 \\ 2 & 1 & 0 & -1 \\ 3 & 2 & 1 & 0 \end{pmatrix}$

(%i3) `C:transpose(A)-A;`

(%o3) $\begin{pmatrix} 0 & 2 & 4 & 6 \\ -2 & 0 & 2 & 4 \\ -4 & -2 & 0 & 2 \\ -6 & -4 & -2 & 0 \end{pmatrix}$

(%i4) `D:transpose(A)+A;`

(%o4) $\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$

Matrica *A* antisimetrična je jer vrijedi $A^{\tau} = -A$.

Primjer 6. Za matricu $A = \begin{pmatrix} 1 & 5 & -2 \\ 2 & 3 & 4 \\ -3 & 0 & 2 \end{pmatrix}$ izračunajte $\det(A)$ i $\det(A^{\tau})$. Što zaključujete?

Rješenje:

```
(%i9) A: matrix(
       [1,5,-2],
       [2,3,4],
       [-3,0,2]
      );
```

$$(\%o9) \quad \begin{pmatrix} 1 & 5 & -2 \\ 2 & 3 & 4 \\ -3 & 0 & 2 \end{pmatrix}$$

```
(%i10)determinant(A);
```

$$(\%o10) \quad -92$$

```
(%i13)B:transpose(A);
```

$$(\%o13) \quad \begin{pmatrix} 1 & 2 & -3 \\ 5 & 3 & 0 \\ -2 & 4 & 2 \end{pmatrix}$$

```
(%i14)determinant(B);
```

$$(\%o14) \quad -92$$

Zaključujemo da je $\det(A)$ = $\det(A^{\tau})$. Jednakost vrijedi općenito.

Primjer 7. Za matrice $A = \begin{pmatrix} 1 & 2 & 1 \\ 2 & 4 & 1 \\ 5 & 2 & 3 \end{pmatrix}$ i $B = \begin{pmatrix} 4 & 1 & 1 \\ -5 & 2 & 0 \\ 1 & 3 & 1 \end{pmatrix}$ izračunajte $\det(A \cdot B)$, $\det(A)$ i $\det(B)$. Što zaključujete?

Rješenje:

```
(%i15) A: matrix([1,2,1], [2,4,1], [5,2,3]);
```

(%o15) $\begin{pmatrix} 1 & 2 & 1 \\ 2 & 4 & 1 \\ 5 & 2 & 3 \end{pmatrix}$

```
(%i16) B: matrix([4,1,1], [-5,2,0], [1,3,1]);
```

(%o16) $\begin{pmatrix} 4 & 1 & 1 \\ -5 & 2 & 0 \\ 1 & 3 & 1 \end{pmatrix}$

```
(%i17) determinant(A.B);
```

(%o17) 32

```
(%i18) determinant(A);
```

(%o18) -8

```
(%i19) determinant(B);
```

(%o19) -4

```
(%i20) determinant(A)·determinant(B);
```

(%o20) 32

Zaključujemo da je $\det\,(A \cdot B) = \det\,(A) \cdot \det\,(B)$. Ova jednakost vrijedi općenito.

Primjer 8. Za matrice $A = \begin{pmatrix} 2 & 3 & 1 \\ 3 & 1 & 3 \\ 3 & 5 & 2 \end{pmatrix}, B = \begin{pmatrix} -7 & 4 \\ 3 & -1 \\ 5 & 3 \end{pmatrix}$ i $C = \begin{pmatrix} 2 & -4 \\ 1 & 1 \\ 5 & 3 \end{pmatrix}$ izračunajte

$A^2 + CB^\tau - 2I.$

Rješenje:

```
(%i6) A: matrix([2,3,1], [3,1,3], [3,5,2]);
```

$$(\%o6)\quad \begin{pmatrix} 2 & 3 & 1 \\ 3 & 1 & 3 \\ 3 & 5 & 2 \end{pmatrix}$$

```
(%i7) B: matrix( [-7,4],  [3,-1],  [5,3]);
```

$$(\%o7)\quad \begin{pmatrix} -7 & 4 \\ 3 & -1 \\ 5 & 3 \end{pmatrix}$$

```
(%i8) C: matrix( [2,-4],  [1,1],  [5,3]);
```

$$(\%o8)\quad \begin{pmatrix} 2 & -4 \\ 1 & 1 \\ 5 & 3 \end{pmatrix}$$

```
(%i9) I:ident(3);
```

$$(\%o9)\quad \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
(%i10)A^^2+C.transpose(B)-2·I;
```

$$(\%o10)\quad \begin{pmatrix} -16 & 24 & 11 \\ 15 & 25 & 20 \\ 4 & 36 & 54 \end{pmatrix}$$

Primjer 9. Zadan je linearan sustav jednadžbi:

$$
\begin{aligned}
2 \cdot x_1 + 2 \cdot x_2 - \ x_3 \ + x_4 &= 4 \\
4 \cdot x_1 + 3 \cdot x_2 - \ x_3 \ + \ 2 \cdot x_4 &= 6 \\
8 \cdot x_1 + 5 \cdot x_2 - 3 \cdot x_3 + 4 \cdot x_4 &= 12 \\
3 \cdot x_1 + 3 \cdot x_2 - 2 \cdot x_3 + 2 \cdot x_4 &= 6
\end{aligned}
$$

Zapišite ga u matričnom obliku, provjerite ima li pripadna matrica sustava inverz te nađite njegovo rješenje.

Rješenje:

```
(%i19)A: matrix( [2,2,-1,1],  [4,3,-1,2],  [8,5,-3,4],  [3,3,-2,2]);
```

(%o19) $\begin{pmatrix} 2 & 2 & -1 & 1 \\ 4 & 3 & -1 & 2 \\ 8 & 5 & -3 & 4 \\ 3 & 3 & -2 & 2 \end{pmatrix}$

```
(%i20)b: matrix( [4],  [6],  [12],  [6]);
```

(%o20) $\begin{pmatrix} 4 \\ 6 \\ 12 \\ 6 \end{pmatrix}$

```
/* Da bi matrica imala inverz, determinanta joj mora
biti različita od 0 */
```

```
(%i21)determinant(A);
```

(%o21) 2

```
/* množimo lijevu i desnu stranu sa inverzom od A */
```

```
(%i22)x:invert(A).b;
```

(%o22) $\begin{pmatrix} 1 \\ 1 \\ -1 \\ -1 \end{pmatrix}$

```
→    /* provjera rješenja */

(%i23) A.x=b;
```

$$(\%o23)\quad \begin{pmatrix} 4 \\ 6 \\ 12 \\ 6 \end{pmatrix} = \begin{pmatrix} 4 \\ 6 \\ 12 \\ 6 \end{pmatrix}$$

13.1.2. Matrične jednadžbe

Primjer 10. Riješite matričnu jednadžbu $XC^{\tau}B - A^3 = 2X + 3A$, pri čemu je

$$A = \begin{pmatrix} 1 & 0 & 3 \\ -3 & 2 & -2 \\ 6 & -1 & 2 \end{pmatrix}, B = \begin{pmatrix} 2 & 3 & -1 \end{pmatrix}, C = \begin{pmatrix} 1 & 4 & 2 \end{pmatrix}$$

Za odgovarajuće matrice dokažite da njihovi inverzi doista postoje, tj. pokažite da su im determinante različite od nule.

Rješenje: Prije unosa u Maximu, moramo riješiti matričnu jednadžbu, odnosno dobiti *X*. Ovdje je jako važno da množenje matrica NIJE KOMUTATIVNO, stoga nije svejedno s koje strane množimo.
$XC^{\tau}B - A^3 = 2X + 3A$
$XC^{\tau}B - 2X = 3A + A^3$ (ne možemo ostaviti samo 2 jer su matrice pa moramo dodati 1)
$X(C^{\tau}B - 2I) = 3A + A^3$ (provjeriti postoji li inverz od $(C^{\tau}B - 2I)$, odnosno je li det$(C^{\tau}B - 2I) \neq 0$)
$X = (3A + A^3)(C^{\tau}B - 2I)$^$(-1)$ (množimo s inverzom s desne strane)

Maxima:

```
(%i4) A: matrix( [1,0,3],   [-3,2,-2],   [6,-1,2]);
```

$$(\%o4)\quad \begin{pmatrix} 1 & 0 & 3 \\ -3 & 2 & -2 \\ 6 & -1 & 2 \end{pmatrix}$$

```
(%i5) B: matrix( [2,3,-1]);
```

$$(\%o5)\quad \begin{pmatrix} 2 & 3 & -1 \end{pmatrix}$$

```
(%i6) C: matrix( [1,4,2]);
```

$$(\%o6)\quad \begin{pmatrix} 1 & 4 & 2 \end{pmatrix}$$

```
(%i8) I:ident(3);
```

$$(\%o8)\quad \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
(%i10) determinant(transpose(C).B-2·I);
(%o10) 40
(%i11) X:(3·A+A^^3).invert(transpose(C).B-2·I);
```

(%o11)
$$\begin{pmatrix} -22 & \frac{153}{4} & -\frac{221}{4} \\ 51 & -\frac{107}{2} & \frac{139}{2} \\ -67 & \frac{253}{4} & -\frac{311}{4} \end{pmatrix}$$

```
→ /* provjera rješenja */
(%i12) X.transpose(C).B-A^^3=2·X+3·A;
```

(%o12)
$$\begin{pmatrix} -41 & \frac{153}{2} & -\frac{203}{2} \\ 93 & -101 & 133 \\ -116 & \frac{247}{2} & -\frac{299}{2} \end{pmatrix} = \begin{pmatrix} -41 & \frac{153}{2} & -\frac{203}{2} \\ 93 & -101 & 133 \\ -116 & \frac{247}{2} & -\frac{299}{2} \end{pmatrix}$$

Primjer 11. Riješite matričnu jednadžbu $ABX + C^2 = B^\tau A^\tau + 2X$, pri čemu je

$$A = \begin{pmatrix} -1 & 2 \\ 2 & -3 \\ 3 & 1 \end{pmatrix}, B = \begin{pmatrix} 1 & 1 & 2 \\ 3 & 1 & 4 \end{pmatrix}, C = \begin{pmatrix} 1 & 1 & 2 \\ 0 & 1 & 0 \\ 1 & 1 & 0 \end{pmatrix}$$

Za odgovarajuće matrice dokažite da njihovi inverzi doista postoje, tj. pokažite da su im determinante različite od nule.

Rješenje: $ABX + C^2 = B^\tau A^\tau + 2X$
$ABX - 2X = B^\tau A^\tau - C^2$
$(AB - 2I)X = B^\tau A^\tau - C^2$ (provjeriti je li det $(AB - 2I) \neq 0$)
$X = (AB - 2I)$^$(-1)(B^\tau A^\tau - C^2)$

Maxima:

(%i2) A: matrix([-1,2], [2,-3], [3,1]);

(%o2) $\begin{pmatrix} -1 & 2 \\ 2 & -3 \\ 3 & 1 \end{pmatrix}$

(%i4) B: matrix([1,1,2], [3,1,4]);

(%o4) $\begin{pmatrix} 1 & 1 & 2 \\ 3 & 1 & 4 \end{pmatrix}$

(%i6) C: matrix([1,1,2], [0,1,0], [1,1,0]);

(%o6) $\begin{pmatrix} 1 & 1 & 2 \\ 0 & 1 & 0 \\ 1 & 1 & 0 \end{pmatrix}$

(%i8) I:ident(3);

(%o8) $\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$

(%i10)determinant(A.B-2·I);

(%o10) -28

(%i11)X:invert(A.B-2·I).(transpose(B).transpose(A)-C^^2);

(%o11) $\begin{pmatrix} -\frac{41}{14} & \frac{55}{7} & -\frac{44}{7} \\ \frac{43}{14} & -\frac{29}{7} & \frac{40}{7} \\ \frac{9}{7} & -\frac{71}{14} & \frac{20}{7} \end{pmatrix}$

Provjera:

```
(%i12) A.B.X+C^^2=transpose(B).transpose(A)+2·X;
```

$$(\%o12)\quad \begin{pmatrix} -\frac{6}{7} & \frac{61}{7} & -\frac{46}{7} \\ \frac{50}{7} & -\frac{65}{7} & \frac{108}{7} \\ \frac{60}{7} & -\frac{127}{7} & \frac{110}{7} \end{pmatrix} = \begin{pmatrix} -\frac{6}{7} & \frac{61}{7} & -\frac{46}{7} \\ \frac{50}{7} & -\frac{65}{7} & \frac{108}{7} \\ \frac{60}{7} & -\frac{127}{7} & \frac{110}{7} \end{pmatrix}$$

13.2. Primjena matrica u ekonomiji

13.2.1. *Input-output* analiza

Primjer 1. Zadana je *input-output* tablica jedne trosektorske ekonomije:

Q_i	Q_{ij}			q_i
Q_1	45	24	42	39
Q_2	30	24	18	48
Q_3	45	30	36	39

Sastavite novu *input-output* tablicu koja odgovara planu proizvodnje: Q_1 = 90, Q_2 = 60, Q_3 = 75 ako je poznato da se tehnološki uvjeti proizvodnje nisu promijenili.

Rješenje: Za teoretski dio rješavanja zadatka pogledati skriptu *Poslovna matematika*, primjer 1.25.

Prvo jednostavno izračunamo vrijednosti Q_i u početnoj tablici.

```
(%i1) Q1:45+24+42+39;
(%o1) 150

(%i2) Q2:30+24+18+48;
(%o2) 120

(%i3) Q3:45+30+36+39;
(%o3) 150

→     /* Nakon toga, odredimo matricu A, prvi stupac dijelimo sa Q1,
      drugi sa Q2 i treći sa Q3 */


(%i4) A: matrix(
       [45/Q1,24/Q2,42/Q3],
       [30/Q1,24/Q2,18/Q3],
       [45/Q1,30/Q2,36/Q3]
      );
```

(%o4) $\begin{pmatrix} \frac{3}{10} & \frac{1}{5} & \frac{7}{25} \\ \frac{1}{5} & \frac{1}{5} & \frac{3}{25} \\ \frac{3}{10} & \frac{1}{4} & \frac{6}{25} \end{pmatrix}$

Sada računamo matricu *T* (*T* = *I* – *A*).

(%i5) I:ident(3);

(%o5) $$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

(%i6) T:I-A;

(%o6) $$\begin{pmatrix} \frac{7}{10} & -\frac{1}{5} & -\frac{7}{25} \\ -\frac{1}{5} & \frac{4}{5} & -\frac{3}{25} \\ -\frac{3}{10} & -\frac{1}{4} & \frac{19}{25} \end{pmatrix}$$

```
→ /*Izračunat ćemo nove q_i-ove pomoću formule q = T Q
  (novi Q-ovi), pa ih prvo unesemo u maximu.
  Nove Q-ovi ćemo označiti sa Qn kako bi se razlikovali od početnih.*/
```

(%i8) Qn: matrix([90], [60], [75]);

(%o8) $$\begin{pmatrix} 90 \\ 60 \\ 75 \end{pmatrix}$$

(%i9) q:T.Qn;

(%o9) $$\begin{pmatrix} 30 \\ 21 \\ 15 \end{pmatrix}$$

```
/*I sad nam još ostaje izgenerirati novu matricu.*/

(%i10) Qmn[i,j]:=A[i,j]·Qn[j,1];
(%o10) Qmn_{i,j}:=A_{i,j} Qn_{j,1}

(%i11) Qmn[i,j]:genmatrix(Qmn,3);
```

$$(\%o11)\quad \begin{pmatrix} 27 & 12 & 21 \\ 18 & 12 & 9 \\ 27 & 15 & 18 \end{pmatrix}$$

Konačno je rješenje nova matrica:

Q_i	Q_{ij}			q_i
90	27	12	21	30
60	18	12	9	21
75	27	15	18	15

Primjer 2. Zadana je *input-output* tablica jedne trosektorske ekonomije:

Q_i	Q_{ij}			q_i
Q_1	30	40	100	130
Q_2	60	120	100	120
Q_3	60	160	150	130

Sastavite novu *input-output* tablicu koja odgovara planu proizvodnje: $q_1 = 143, q_2 = 132$ i $q_3 = 143$ ako je poznato da se tehnološki uvjeti proizvodnje nisu promijenili.

Rješenje: Prvo izračunamo parametre koji nam nedostaju u početnoj tablici.

```
(%i1) Q1:30+40+100+130;
(%o1) 300

(%i2) Q2:60+120+100+120;
(%o2) 400

(%i3) Q3:60+160+150+130;
(%o3) 500
```

Sada izračunamo matrice *A* i *T*.

```
(%i4) A: matrix(
       [30/Q1,40/Q2,100/Q3],
       [60/Q1,120/Q2,100/Q3],
       [60/Q1,160/Q2,150/Q3]
      );
```

(%o4) $$\begin{pmatrix} \frac{1}{10} & \frac{1}{10} & \frac{1}{5} \\ \frac{1}{5} & \frac{3}{10} & \frac{1}{5} \\ \frac{1}{5} & \frac{2}{5} & \frac{3}{10} \end{pmatrix}$$

```
(%i5) I:ident(3);
```

(%o5) $$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
(%i6) T:I-A;
```

(%o6) $$\begin{pmatrix} \frac{9}{10} & -\frac{1}{10} & -\frac{1}{5} \\ -\frac{1}{5} & \frac{7}{10} & -\frac{1}{5} \\ -\frac{1}{5} & -\frac{2}{5} & \frac{7}{10} \end{pmatrix}$$

Iz formule *q* = *TQ* izračunamo *Q* kao *T*^(–1).*q*.

```
(%i9) qn: matrix([143], [132], [143]);
```

(%o9) $$\begin{pmatrix} 143 \\ 132 \\ 143 \end{pmatrix}$$

```
→    /* Budući nam treba inverzna matrica matrice T, moramo
     provjeriti da je det(T≠0 */)

(%i8) determinant(T);
```

(%o8) $\frac{307}{1000}$

```
(%i11)Qn:invert(T).qn;
```

(%o11) $\begin{pmatrix} 330 \\ 440 \\ 550 \end{pmatrix}$

```
(%i12)Qmn[i,j]:=A[i,j]·Qn[j,1];
```

(%o12) $Qmn_{i,j} := A_{i,j}\, Qn_{j,1}$

```
(%i13)Qmn[i,j]:genmatrix(Qmn,3);
```

(%o13) $\begin{pmatrix} 33 & 44 & 110 \\ 66 & 132 & 110 \\ 66 & 176 & 165 \end{pmatrix}$

Konačno je rješenje nova matrica:

Q_i	Q_{ij}			q_i
330	33	44	110	143
440	66	132	110	132
550	66	176	165	143

Primjer 3. Zadana je *input-output* tablica jedne trosektorske ekonomije:

Q_i	Q_{ij}			q_i
100	26	15	32	q_1
Q_2	40	12	12	11
Q_3	34	21	24	21

Sastavite novu *input-output* tablicu koja odgovara planu proizvodnje: $Q_2 = 75, Q_3 = 125$ i $q_1 = 19$ ako je poznato da se tehnološki uvjeti proizvodnje nisu promijenili.

Rješenje: Prvo izračunamo parametre koji nam nedostaju u početnoj tablici.

```
(%i1) q1:100-(26+15+32);
(%o1) 27

(%i2) Q2:40+12+12+11;
(%o2) 75

(%i3) Q3:34+21+24+21;
(%o3) 100

 →    /* Moramo unijeti i Q1 kako bi maxima znala dalje računati) */

(%i4) Q1:100;
(%o4) 100

 →    /* Sada računamo matricu A i T */

(%i5) A: matrix(
       [26/Q1,15/Q2,32/Q3],
       [40/Q1,12/Q2,12/Q3],
       [34/Q1,21/Q2,24/Q3]
      );
```

$$(\%o5)\quad \begin{pmatrix} \frac{13}{50} & \frac{1}{5} & \frac{8}{25} \\ \frac{2}{5} & \frac{4}{25} & \frac{3}{25} \\ \frac{17}{50} & \frac{7}{25} & \frac{6}{25} \end{pmatrix}$$

```
(%i6) I:ident(3);
```

$$(\%o6) \quad \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
(%i7) T:I-A;
```



```
(%i7) T:I-A;
```


$$(\%o7) \quad \begin{pmatrix} \frac{37}{50} & -\frac{1}{5} & -\frac{8}{25} \\ -\frac{2}{5} & \frac{21}{25} & -\frac{3}{25} \\ -\frac{17}{50} & -\frac{7}{25} & \frac{19}{25} \end{pmatrix}$$

```
(%i8) Qn: matrix([Qn1], [75], [125]);
```

$$(\%o8) \quad \begin{pmatrix} Qn1 \\ 75 \\ 125 \end{pmatrix}$$

```
(%i9) q:T.Qn;
```

$$(\%o9) \quad \begin{pmatrix} \frac{37\,Qn1}{50} - 55 \\ 48 - \frac{2\,Qn1}{5} \\ 74 - \frac{17\,Qn1}{50} \end{pmatrix}$$

Budući da nam je u zadatku zadano $q_1 = 19$, rješavanjem prve jednadžbe dobit ćemo Q_{n1}.

```
(%i10)solve([(37·Qn1)/50-55=19],[Qn1]);
```

$$(\%o10) \quad [Qn1 = 100]$$

Sada ponovimo prethodna dva koraka, odnosno unesemo sve vrijednosti Q_n i dobit ćemo nove vrijednosti q.

```
(%i11) Qn: matrix([100], [75], [125]);
```

$$(\%o11)\quad \begin{pmatrix} 100 \\ 75 \\ 125 \end{pmatrix}$$

```
(%i12) q:T.Qn;
```

$$(\%o12)\quad \begin{pmatrix} 19 \\ 8 \\ 40 \end{pmatrix}$$

```
→ /* Ostaje nam još izgenerirati novu matricu. */
```

```
(%i13) Qmn[i,j]:=A[i,j]·Qn[j,1];
```

$$(\%o13)\quad Qmn_{i,j} := A_{i,j}\, Qn_{j,1}$$

```
(%i14) Qmn[i,j]:genmatrix(Qmn,3);
```

$$(\%o14)\quad \begin{pmatrix} 26 & 15 & 40 \\ 40 & 12 & 15 \\ 34 & 21 & 30 \end{pmatrix}$$

Konačno je rješenje nova matrica:

Q_i	Q_{ij}			q_i
100	26	15	40	19
75	40	12	15	8
125	34	21	30	40

13.3. Limesi, derivacije, integrali

Problem izračuna limesa, derivacija ili integrala u Maximi svodi se na ispravno upisivanje odgovarajuće funkcije.

Primjer 1. Izračunajte limese sljedećih funkcija:

a) $\lim\limits_{x \to 1} \frac{x^2-3x+2}{1-x^3}$

b) $\lim\limits_{x \to 0} \frac{\sin(\sqrt[3]{x+8}-2)}{\sin 3x}$

c) $\lim\limits_{x \to 0} \frac{\ln(1-sin^2 x)}{1-\cos x}$

Rješenje:

```
→   /* a) */

(%i1) f(x):=(x^2-3·x+2)/(1-x^3);
```

(%o1) $f(x) := \frac{x^2-3x+2}{1-x^3}$

```
(%i2) limit(f(x),x,1);
```

(%o2) $\frac{1}{3}$

```
→   /* b) */

(%i3) g(x):=sin((x+8)^(1/3)-2)/sin(3·x);
```

(%o3) $g(x) := \frac{\sin\left((x+8)^{1/3}-2\right)}{\sin(3x)}$

```
(%i4) limit(g(x),x,0);
```

(%o4) $-\frac{-\sin(2)^2-\cos(2)^2}{36}$

```
(%i5) float(%);
(%o5) 0.02777777777777778
```

```
→    /* c) */

(%i6) h(x):=log(1-sin(x)^2)/(1-cos(x));
```

$$(\%o6)\quad h(x) := \frac{\log\left(1-\sin(x)^2\right)}{1-\cos(x)}$$

```
(%i8) limit(h(x),x,0);
(%o8) -2
```

Primjer 2. Izračunajte derivacije sljedećih funkcija:

a) $f(x) = (3 - 2\sin x)^5$

b) $f(x) = \frac{2}{3}\sqrt[5]{x^3} + 5\ln 12 + 3\ln x$

c) $f(x) = \frac{x^2+3x}{e^{x^2+4x}}$

d) $f(x) = \sqrt[3]{2e^x - 2^x + 1} + ln^5 x$

Rješenje:

```
→    /* a) */

(%i9) f(x):=(3-2·sin(x))^5;
```

$$(\%o9)\quad f(x) := (3-2\sin(x))^5$$

```
(%i10) diff(f(x),x);
```

$$(\%o10)\quad -10\cos(x)\,(3-2\sin(x))^4$$

→ /* b) */

```
(%i11) f(x):=2/3·x^(3/5)+5·log(12)+3·log(x);
```

(%o11) $f(x) := \frac{2}{3} x^{3/5} + 5 \log(12) + 3 \log(x)$

```
(%i12) diff(f(x),x);
```

(%o12) $\frac{2}{5 x^{2/5}} + \frac{3}{x}$

→ /* c) */

```
(%i13) f(x):=(x^2+3·x)/(%e^(x^2+4·x));
```

(%o13) $f(x) := \frac{x^2 + 3 x}{\%e^{x^2 + 4 x}}$

```
(%i14) diff(f(x),x);
```

(%o14) $(-2 x - 4)(x^2 + 3 x) \%e^{-x^2 - 4 x} + (2 x + 3) \%e^{-x^2 - 4 x}$

→ /* d) */

```
(%i15) f(x):=(2·%e^x-2^x+1)^(1/3)+log(x)^5;
```

(%o15) $f(x) := (2 \%e^x - 2^x + 1)^{1/3} + \log(x)^5$

```
(%i16) diff(f(x),x);
```

(%o16) $\frac{5 \log(x)^4}{x} + \frac{2 \%e^x - \log(2) 2^x}{3 (2 \%e^x - 2^x + 1)^{2/3}}$

Primjer 3. Izračunajte sljedeće neodređene integrale:

a) $\int (x^2-3)\sqrt[5]{(x^3-9x)}dx$

b) $\int x\,(3x^2+1)^{10}dx$

c) $\int \frac{x}{\sqrt{(x^2+1)^3}}dx$

Rješenje:

```
→   /* a) */
(%i4) f(x):=(x^2-3)·(x^3-9·x)^(1/5);
```

(%o4) $f(x):=(x^2-3)\,(x^3-9x)^{1/5}$

```
(%i5) integrate(f(x),x);
```

(%o5) $\frac{5\,(x^3-9x)^{6/5}}{18}$

```
→   /* b) */
(%i2) f(x):=x·(3·x^2+1)^10;
```

(%o2) $f(x):=x\,(3x^2+1)^{10}$

```
(%i3) integrate(f(x),x);
```

(%o3) $\frac{(3x^2+1)^{11}}{66}$

```
→   /* c) */
(%i7) f(x):=x/(x^2+1)^(1/3);
```

(%o7) $f(x):=\frac{x}{(x^2+1)^{1/3}}$

```
(%i8) integrate(f(x),x);
```

(%o8) $\frac{3\,(x^2+1)^{2/3}}{4}$

Primjer 4. Izračunajte sljedeće određene integrale:

a) $\int_0^8 (1+\sqrt{2x}+\sqrt[3]{x})\,dx$

b) $\int_1^4 \frac{1+\sqrt{x}}{x^2}dx$

c) $\int_{-\pi}^{\pi} x\,(\sin x - 1)\;dx$

Rješenje:

```
→      /* a) */

(%i1)  f(x):=(1+(2·x)^(1/2)+x^(1/3));
```

(%o1) $f(x):=1+(2\,x)^{1/2}+x^{1/3}$

```
(%i2)  integrate(f(x),x,0,8);
```

(%o2) $\frac{124}{3}$

```
→      /* b) */

(%i4)  f(x):=(1+x^(1/2))/x^2;
```

(%o4) $f(x):=\frac{1+x^{1/2}}{x^2}$

```
(%i5)  integrate(f(x),x,1,4);
```

(%o5) $\frac{7}{4}$

```
→      /* c) */

(%i6)  f(x):=x·(sin(x)-1);
```

(%o6) $f(x):=x\,(\sin(x)-1)$

```
(%i7) integrate(f(x),x,-%pi,%pi);
```

$$(\%o7)\quad \frac{\pi^2+2\pi}{2}-\frac{\pi^2-2\pi}{2}$$

```
(%i8) float(%);
(%o8) 6.283185307179586
```

13.3.1. Primjena derivacija i integrala u ekonomiji

Primjer 5. Zadana je funkcija graničnih troškova:

$$t(Q)=\ln(Q+4)$$

Odredite funkciju ukupnih troškova T ako je poznato da fiksni troškovi iznose $8\cdot\ln(2)$.

Rješenje:

```
(%i1) t(Q):=log(Q+1);
(%o1) t(Q):=log(Q+1)

(%i2) integrate(t(Q),Q);
(%o2) (Q+1) log(Q+1)-Q-1

(%i3) T(Q):=(Q+1)·log(Q+1)-Q-1+C;
(%o3) T(Q):=(Q+1) log(Q+1)-Q-1+C

(%i4) solve(T(0)=8·log(2),C);
(%o4) [C=8 log(2)+1]
```

Primjer 6. Zadana je funkcija graničnih troškova:

$$t(Q) = \frac{e^{2\sqrt{Q}}}{\sqrt{Q}}$$

Odredite funkciju ukupnih troškova T ako je poznato da fiksni troškovi iznose 1.

Rješenje:

```
(%i1) t(Q):=(%e^(2·Q^(1/2)))/Q^(1/2);
```

$$(\%o1)\quad t(Q) := \frac{\%e^{2\,Q^{1/2}}}{Q^{1/2}}$$

```
(%i2) integrate(t(Q),Q);
```

$$(\%o2)\quad \%e^{2\sqrt{Q}}$$

```
(%i3) T(Q):=%e^(2·sqrt(Q))+C;
```

$$(\%o3)\quad T(Q) := \%e^{2\sqrt{Q}} + C$$

```
(%i4) solve(T(0)=1,C);
```

$$(\%o4)\quad [C=0]$$

Primjer 7. Zadana je funkcija ukupnih troškova:

$$T(Q) = \frac{4}{5}\sqrt[3]{Q^4} + 3\ln(e+Q) + 9$$

Odredite funkciju graničnih troškova t. Koliko iznose fiksni troškovi?

Rješenje:

```
(%i1) T(Q):=4/5·Q^(4/3)+3·log(%e+Q)+9;
```

(%o1) $T(Q) := \frac{4}{5} Q^{4/3} + 3\log(\%e + Q) + 9$

```
→  /* fiksni troškovi */

(%i2) T(0);
(%o2) 12

(%i3) diff(T(Q),Q);
```

(%o3) $\frac{3}{Q+\%e} + \frac{16\,Q^{1/3}}{15}$

```
(%i4) t(Q):=3/(Q+%e)+(16·Q^(1/3))/15;
```

(%o4) $t(Q) := \frac{3}{Q+\%e} + \frac{16\,Q^{1/3}}{15}$

13.4. Površine

Primjer 1. Izračunajte površinu lika omeđenog krivuljama $y = \frac{x^2}{2}$ i $y = x + \frac{3}{2}$. Nacrtajte sliku.

Rješenje:

```
(%i1) f(x):=x^2/2;
```

(%o1) $f(x) := \frac{x^2}{2}$

```
(%i2) g(x):=x+3/2;
```

(%o2) $g(x) := x + \frac{3}{2}$

```
→ /* točke presjeke pravca i parabole */

(%i3) solve([f(x)=g(x)],[x]);
```

(%o3) $[x = 3, x = -1]$

```
(%i8) wxplot2d([f(x),g(x)],[x,-2,4]);
(%t8)
```




```
→ /* Površina */

(%i6) integrate(g(x)-f(x),x,-1,3);
```

(%o6) $\frac{16}{3}$

Primjer 2. Izračunajte površinu lika omeđenog krivuljama $y = -2x^2 + 3x + 1$ i $y = x - 3$. Nacrtajte sliku.

Rješenje:

```
(%i18) f(x):=-2·x^2+3·x+1;
(%o18) f(x):=(-2)x^2+3x+1

(%i19) g(x):=x-3;
(%o19) g(x):=x-3

 →    /* presjek pravca i parabole */

(%i20) solve([f(x)=g(x)],[x]);
(%o20) [x=2,x=-1]

(%i21) wxplot2d([f(x),g(x)],[x,-2,3]);
(%t21)
```




```
 →    /* Površina */

(%i22) integrate(f(x)-g(x),x,-1,2);
(%o22) 9
```

13.5. Redovi

Ovdje ćemo navesti tri osnovna kriterija za konvergenciju redova koje primjenjujemo u primjerima navedenima u nastavku.

Poredbeni kriterij (granični oblik): Neka su (a_n) i (b_n) nizovi s pozitivnim članovima tako da postoji

$$\lim_{n\to\infty} \frac{a_n}{b_n} = L \in R^+$$

Tada redovi $\sum_{n=1}^{\infty} a_n$ i $\sum_{n=1}^{\infty} b_n$ ili oba konvergiraju ili oba divergiraju.

Napomena: Red $\sum_{n=1}^{\infty} \frac{1}{n^r}$ konvergira za $r > 1$ i divergira za $r \leq 1$.

D'Alambertov kriterij: Neka je $\sum_{n=1}^{\infty} a_n$ red s pozitivnim članovima. Pretpostavimo da je

$$q \coloneqq \lim_{n\to\infty} \frac{a_{n+1}}{a_n} \in R$$

Tada vrijedi:

i) ako je $q < 1$, red $\sum_{n=1}^{\infty} a_n$ je konvergentan
ii) ako je $q > 1$, red $\sum_{n=1}^{\infty} a_n$ je divergentan
iii) ako je $q = 1$, nema odluke o konvergenciji reda $\sum_{n=1}^{\infty} a_n$

Cauchyjev kriterij: Neka je $\sum_{n=1}^{\infty} a_n$ red s pozitivnim članovima. Pretpostavimo da je

$$q \coloneqq \lim_{n\to\infty} \sqrt[n]{a_n} \in R$$

Tada vrijedi:

i) ako je $q < 1$, red $\sum_{n=1}^{\infty} a_n$ je konvergentan
ii) ako je $q > 1$, red $\sum_{n=1}^{\infty} a_n$ je divergentan
iii) ako je $q = 1$, nema odluke o konvergenciji reda $\sum_{n=1}^{\infty} a_n$

Primjer 1. Ispitajte konvergenciju navedenog reda. Ako je red konvergentan, procijenite njegovu sumu.

$$\sum_{n=1}^{\infty} \frac{3}{\sqrt[3]{2n\sqrt{(4n-1)}\,(2n+3)}}$$

Rješenje: Ovaj ćemo zadatak riješiti poredbenim kriterijem. Prvo je potrebno naći najveću potenciju ovog reda, a ona iznosi:

$$(n \cdot n^{\frac{1}{2}} \cdot n)^{\frac{1}{3}} = n^{\frac{5}{6}}$$

Naš je red s kojim ćemo usporediti početni:

$$\sum_{n=1}^{\infty} \frac{1}{n^{\frac{5}{6}}}$$

Za njega znamo (vidi prethodnu napomenu) da je divergentan jer je $\frac{5}{6} < 1$. Budući da postoji $\lim_{n\to\infty} \frac{a_n}{b_n}$, možemo zaključiti da je i naš početni red divergentan.

```
(%i1) a(n):=3/(2·n·(4·n-1)^(1/2)·(2·n+3))^(1/3);
```

$$\text{(\%o1)}\quad a(n) := \frac{3}{\left(2\,n\,(4\,n-1)^{1/2}\,(2\,n+3)\right)^{1/3}}$$

```
→     /* nužan uvjet konvergencije */;

(%i2) limit(a(n),n,inf);
(%o2) 0

(%i3) b(n):=1/n^(5/6);
```

$$\text{(\%o3)}\quad b(n) := \frac{1}{n^{5/6}}$$

```
→     /* poredbeni kriterij */

(%i4) limit(a(n)/b(n),n,inf);
```

$$\text{(\%o4)}\quad \frac{3}{2}$$

Primjer 2. Ispitajte konvergenciju navedenog reda. Ako je red konvergentan, procijenite njegovu sumu.

$$\sum_{n=1}^{\infty} \frac{3 \cdot 8 \cdot \,\cdots\, \cdot (5n-2)}{4 \cdot 7 \cdot \cdots \, \cdot (3n+1)}$$

Rješenje: Ovaj ćemo zadatak riješiti D'Alambertovim kriterijem. Promatrat ćemo $\lim\limits_{n\to\infty} \frac{a_{n+1}}{a_n}$

$$\lim_{n\to\infty} \frac{\dfrac{3\cdot 8\cdot \cdots \cdot(5n-2)\cdot(5n+3)}{4\cdot 7\cdot \cdots \cdot(3n+1)\cdot(3n+4)}}{\dfrac{3\cdot 8\cdot \cdots \cdot(5n-2)}{4\cdot 7\cdot \cdots \cdot(3n+1)}} = \lim_{n\to\infty} \frac{5n+3}{3n+4} = \frac{5}{3} > 1$$

Budući da je $\frac{5}{3} > 1$, prema D'Alambertovu kriteriju zaključujemo da je početni red divergentan.

```
(%i10) limit((5·n+3)/(3·n+4),n,inf);
(%o10)  5/3
```

Primjer 3. Ispitajte konvergenciju navedenog reda. Ako je red konvergentan, procijenite njegovu sumu.

$$\sum_{n=1}^{\infty} \left(\frac{2n^2+1}{5n^2+3n}\right)^n$$

Rješenje: Ovaj ćemo zadatak riješiti Cauchyjevim kriterijem. Promatrat ćemo $\lim\limits_{n\to\infty} \sqrt[n]{a_n}$.

$$\lim_{n\to\infty} \sqrt[n]{\left(\frac{2n^2+1}{5n^2+3n}\right)^n} = \lim_{n\to\infty} \frac{2n^2+1}{5n^2+3n} = \frac{2}{5} < 1$$

Budući da je $\frac{2}{5} < 1$, prema Cauchyjevu kriteriju zaključujemo da je početni red konvergentan.

```
(%i1) a(n):=((2·n^2+1)/(5·n^2+3·n))^n;
(%o1) a(n):=((2 n^2 +1)/(5 n^2 +3 n))^n

(%i3) limit(a(n)^(1/n),n,inf);
(%o3)  2/5
```

Sada ćemo procijeniti sumu reda.

```
(%i10) S:0;
(%o10) 0

(%i11) for n:1 thru 300 do
           S:S+a(n);
(%o11) done

 →      /* do 100 */

(%i9) float(S);
(%o9) 0.5654999390323234

 →      /* do 300 */

(%i12) float(S);
(%o12) 0.5654999390323234
```

Primjer 4. Ispitajte konvergenciju navedenog reda. Ako je red konvergentan, procijenite njegovu sumu.

$$\sum_{n=1}^{\infty} \left(\frac{n}{3n+1}\right)^{2n+1}$$

Rješenje: Ovaj ćemo zadatak riješiti Cauchyjevim kriterijem. Promatrat ćemo $\lim_{n\to\infty} \sqrt[n]{a_n}$.

$$\lim_{n\to\infty} \sqrt[n]{\left(\frac{n}{3n+1}\right)^{2n+1}} = \frac{1}{9} < 1$$

Budući da je $\frac{1}{9} < 1$, prema Cauchyjevu kriteriju zaključujemo da je početni red konvergentan.

```
(%i14) a(n):=(n/(3·n+1))^(2·n+1);
```

(%o14) $a(n):=\left(\frac{n}{3n+1}\right)^{2n+1}$

```
(%i15) limit(a(n),n,inf);
(%o15) 0

(%i16) limit(a(n)^(1/n),n,inf);
```

(%o16) $\frac{1}{9}$

Sada ćemo procijeniti sumu reda.

```
(%i23) S:0;
(%o23) 0

(%i24) for n:1 thru 300 do
           S:S+a(n);
(%o24) done

(%i26) float(S); /* do 100 */
(%o26) 0.01777551383013738

 →     float(S); /* do 300 */
(%o25) 0.01777551383013738
```

Popis slika

Slika 1. Sučelje WxMaxima ... 2
Slika 2. Primjer pridruživanja u Maximi ... 3
Slika 3. Primjer poziva pomoći u Maximi ... 4
Slika 4. Primjer ugrađenih konstanti u Maximi ... 5
Slika 5. Primjeri algebarskih operacija u Maximi ... 6
Slika 6. Primjer rezultata u Maximi ... 7
Slika 7. Primjer logičkih operatora ... 8
Slika 8. Primjeri funkcija u Maximi ... 9
Slika 9. Primjer definiranja funkcija ... 10
Slika 10. Primjer uvjetno definirane funkcije ... 10
Slika 11. Primjer manipulacije izrazima ... 11
Slika 12. Primjer računanja limesa ... 12
Slika 13. Primjer derivacija ... 13
Slika 14. Primjer računanja integrala ... 13
Slika 15. Primjer sume i produkta prvih n brojeva ... 14
Slika 16. Primjer crtanja dvodimenzionalnih grafova ... 16
Slika 17. Primjer trodimenzionalnog grafa ... 16
Slika 18. Primjer dodatnih opcija za grafički prikaz ... 17
Slika 19. Primjer unosa matrice ... 19
Slika 20. Primjer operacija s matricama ... 20
Slika 21. Primjer rada s matricama ... 21
Slika 22. Primjer generiranja matrice ... 22
Slika 23. Primjer rješavanja linearnih jednadžbi ... 23
Slika 24. Primjer rješavanja jednadžbi ... 24
Slika 25. Primjer rješavanja nejednadžbi ... 24
Slika 26. Primjer *for* petlje ... 25

Literatura

1. The Computer Algebra Program Maxima – a Tutorial, dostupno na: https://maxima.sourceforge.io/docs/tutorial/en/gaertner-tutorial-revision/Contents.htm (pristupljeno 17. ožujka 2021.).
2. Lopatič, J. Poslovna matematika (2015). Zaprešić: Veleučilište s pravom javnosti Baltazar Zaprešić